pragma solidity ^0.4.25;




/**
* @title AccessControl
* @notice This contract defines organizational roles and permissions.
*/
contract AccessControl {
    /**
     * @notice ContractUpgrade is the event that will be emitted if we set a new contract address
     */
    event ContractUpgrade(address newContract);
    event Paused();
    event Unpaused();

    /**
     * @notice CEO's address FOOBAR
     */
    address public ceoAddress;

    /**
     * @notice CFO's address
     */
    address public cfoAddress;

    /**
     * @notice COO's address
     */
    address public cooAddress;

    /**
     * @notice withdrawal address
     */
    address public withdrawalAddress;

    bool public paused = false;

    /**
     * @dev Modifier to make a function only callable by the CEO
     */
    modifier onlyCEO() {
        require(msg.sender == ceoAddress);
        _;
    }

    /**
     * @dev Modifier to make a function only callable by the CFO
     */
    modifier onlyCFO() {
        require(msg.sender == cfoAddress);
        _;
    }

    /**
     * @dev Modifier to make a function only callable by the COO
     */
    modifier onlyCOO() {
        require(msg.sender == cooAddress);
        _;
    }

    /**
     * @dev Modifier to make a function only callable by C-level execs
     */
    modifier onlyCLevel() {
        require(
        msg.sender == cooAddress ||
        msg.sender == ceoAddress ||
        msg.sender == cfoAddress
        );
        _;
    }

    /**
     * @dev Modifier to make a function only callable by CEO or CFO
     */
    modifier onlyCEOOrCFO() {
        require(
        msg.sender == cfoAddress ||
        msg.sender == ceoAddress
        );
        _;
    }

    /**
     * @dev Modifier to make a function only callable by CEO or COO
     */
    modifier onlyCEOOrCOO() {
        require(
        msg.sender == cooAddress ||
        msg.sender == ceoAddress
        );
        _;
    }

    /**
     * @notice Sets a new CEO
     * @param _newCEO - the address of the new CEO
     */
    function setCEO(address _newCEO) external onlyCEO {
        require(_newCEO != address(0));
        ceoAddress = _newCEO;
    }

    /**
     * @notice Sets a new CFO
     * @param _newCFO - the address of the new CFO
     */
    function setCFO(address _newCFO) external onlyCEO {
        require(_newCFO != address(0));
        cfoAddress = _newCFO;
    }

    /**
     * @notice Sets a new COO
     * @param _newCOO - the address of the new COO
     */
    function setCOO(address _newCOO) external onlyCEO {
        require(_newCOO != address(0));
        cooAddress = _newCOO;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     */
    modifier whenNotPaused() {
        require(!paused);
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     */
    modifier whenPaused() {
        require(paused);
        _;
    }

    /**
     * @notice called by any C-level to pause, triggers stopped state
     */
    function pause() public onlyCLevel whenNotPaused {
        paused = true;
        emit Paused();
    }

    /**
     * @notice called by the CEO to unpause, returns to normal state
     */
    function unpause() public onlyCEO whenPaused {
        paused = false;
        emit Unpaused();
    }
}

/**
 * @title ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/20
 */
interface IERC20 {
  function totalSupply() external view returns (uint256);

  function balanceOf(address who) external view returns (uint256);

  function allowance(address owner, address spender)
    external view returns (uint256);

  function transfer(address to, uint256 value) external returns (bool);

  function approve(address spender, uint256 value)
    external returns (bool);

  function transferFrom(address from, address to, uint256 value)
    external returns (bool);

  event Transfer(
    address indexed from,
    address indexed to,
    uint256 value
  );

  event Approval(
    address indexed owner,
    address indexed spender,
    uint256 value
  );
}


/**
 * @title SafeMath
 * @dev Math operations with safety checks that revert on error
 */
library SafeMath {

  /**
  * @dev Multiplies two numbers, reverts on overflow.
  */
  function mul(uint256 a, uint256 b) internal pure returns (uint256) {
    // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
    // benefit is lost if 'b' is also tested.
    // See: https://github.com/OpenZeppelin/openzeppelin-solidity/pull/522
    if (a == 0) {
      return 0;
    }

    uint256 c = a * b;
    require(c / a == b);

    return c;
  }

  /**
  * @dev Integer division of two numbers truncating the quotient, reverts on division by zero.
  */
  function div(uint256 a, uint256 b) internal pure returns (uint256) {
    require(b > 0); // Solidity only automatically asserts when dividing by 0
    uint256 c = a / b;
    // assert(a == b * c + a % b); // There is no case in which this doesn't hold

    return c;
  }

  /**
  * @dev Subtracts two numbers, reverts on overflow (i.e. if subtrahend is greater than minuend).
  */
  function sub(uint256 a, uint256 b) internal pure returns (uint256) {
    require(b <= a);
    uint256 c = a - b;

    return c;
  }

  /**
  * @dev Adds two numbers, reverts on overflow.
  */
  function add(uint256 a, uint256 b) internal pure returns (uint256) {
    uint256 c = a + b;
    require(c >= a);

    return c;
  }

  /**
  * @dev Divides two numbers and returns the remainder (unsigned integer modulo),
  * reverts when dividing by zero.
  */
  function mod(uint256 a, uint256 b) internal pure returns (uint256) {
    require(b != 0);
    return a % b;
  }
}


contract LockToken is AccessControl {
    mapping (address => uint256) private lockTokenNum;
    mapping (address => uint256) private lockTokenTime;
    
    event SetLockTokenNum(address from,uint256 num);
    event SetLockTokenTime(address from,uint256 time);
    event SetLockTokenInfo(address from,uint256 num,uint256 time);

    function setLockTokenNum (address from,uint256 num) public  whenNotPaused onlyCEO {
        require(from != address(0));
        lockTokenNum[from] = num;
        emit SetLockTokenNum(from,num);
    }
    
    function setLockTokenTime(address from,uint256 time) public whenNotPaused onlyCEO {
        require(from != address(0));
        lockTokenTime[from] = time;
        emit SetLockTokenTime(from,time);
    }
    
    function setLockTokenInfo(address from,uint256 num,uint256 time) public whenNotPaused onlyCEO {
        require(from != address(0));
        lockTokenNum[from] = num;
        lockTokenTime[from] = time;
        emit SetLockTokenInfo(from,num,time);
    }
    
    
    function setLockTokenInfoList (address[] froms,uint256[] nums, uint256[] times) public whenNotPaused onlyCEO {
        for(uint256 i =0;i<froms.length ;i++ ){
            require(froms[i] != address(0));
            lockTokenNum[froms[i]] = nums[i];
            lockTokenTime[froms[i]] = times[i];
        }
    }
    
    function getLockTokenNum (address from) public view returns (uint256) {
        require(from != address(0));
        return lockTokenNum[from];
    }
    
    function getLockTokenTime(address from) public view returns (uint256) {
        require(from != address(0));
        return lockTokenTime[from];
    }
    
    function getBlockTime() public view returns(uint256){
        return block.timestamp;
    }
}

/**
 * @title Standard ERC20 token
 *
 * @dev Implementation of the basic standard token.
 * https://github.com/ethereum/EIPs/blob/master/EIPS/eip-20.md
 * Originally based on code by FirstBlood: https://github.com/Firstbloodio/token/blob/master/smart_contract/FirstBloodToken.sol
 */
contract ERC20 is IERC20,LockToken{
  using SafeMath for uint256;

  mapping (address => uint256) public _balances;

  mapping (address => mapping (address => uint256)) public _allowed;

  uint256 public _totalSupply;

  /**
  * @dev Total number of tokens in existence
  */
  function totalSupply() public view returns (uint256) {
    return _totalSupply;
  }

  /**
  * @dev Gets the balance of the specified address.
  * @param owner The address to query the balance of.
  * @return An uint256 representing the amount owned by the passed address.
  */
  function balanceOf(address owner) public view returns (uint256) {
    return _balances[owner];
  }

  /**
   * @dev Function to check the amount of tokens that an owner allowed to a spender.
   * @param owner address The address which owns the funds.
   * @param spender address The address which will spend the funds.
   * @return A uint256 specifying the amount of tokens still available for the spender.
   */
  function allowance(
    address owner,
    address spender
   )
    public
    view whenNotPaused
    returns (uint256) 
  {
    return _allowed[owner][spender];
  }

  /**
  * @dev Transfer token for a specified address
  * @param to The address to transfer to.
  * @param value The amount to be transferred.
  */
  function transfer(address to, uint256 value) public whenNotPaused returns (bool) {
    require(value <= _balances[msg.sender]);
    require(to != address(0));
    uint256 time = getLockTokenTime(msg.sender);
    uint256 blockTime = block.timestamp;
    require(blockTime >time);
    _balances[msg.sender] = _balances[msg.sender].sub(value);
    _balances[to] = _balances[to].add(value);
    emit Transfer(msg.sender, to, value);
    return true;
  }

  /**
   * @dev Approve the passed address to spend the specified amount of tokens on behalf of msg.sender.
   * Beware that changing an allowance with this method brings the risk that someone may use both the old
   * and the new allowance by unfortunate transaction ordering. One possible solution to mitigate this
   * race condition is to first reduce the spender's allowance to 0 and set the desired value afterwards:
   * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
   * @param spender The address which will spend the funds.
   * @param value The amount of tokens to be spent.
   */
  function approve(address spender, uint256 value) public whenNotPaused returns (bool) {
    require(spender != address(0));
    uint256 time = getLockTokenTime(msg.sender);
    uint256 blockTime = block.timestamp;
    require(blockTime >time);
    _allowed[msg.sender][spender] = value;
    emit Approval(msg.sender, spender, value);
    return true;
  }

  /**
   * @dev Transfer tokens from one address to another
   * @param from address The address which you want to send tokens from
   * @param to address The address which you want to transfer to
   * @param value uint256 the amount of tokens to be transferred
   */
  function transferFrom(
    address from,
    address to,
    uint256 value
  )
    public whenNotPaused
    returns (bool)
  {
    require(value <= _balances[from]);
    require(value <= _allowed[from][msg.sender]);
    require(to != address(0));
    uint256 time = getLockTokenTime(from);
    uint256 blockTime = block.timestamp;
    require(blockTime >time);
    _balances[from] = _balances[from].sub(value);
    _balances[to] = _balances[to].add(value);
    _allowed[from][msg.sender] = _allowed[from][msg.sender].sub(value);
    emit Transfer(from, to, value);
    return true;
  }

  /**
   * @dev Increase the amount of tokens that an owner allowed to a spender.
   * approve should be called when allowed_[_spender] == 0. To increment
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   * @param spender The address which will spend the funds.
   * @param addedValue The amount of tokens to increase the allowance by.
   */
  function increaseAllowance(
    address spender,
    uint256 addedValue
  )
    public
    returns (bool)
  {
    require(spender != address(0));

    _allowed[msg.sender][spender] = (
      _allowed[msg.sender][spender].add(addedValue));
    emit Approval(msg.sender, spender, _allowed[msg.sender][spender]);
    return true;
  }

  /**
   * @dev Decrease the amount of tokens that an owner allowed to a spender.
   * approve should be called when allowed_[_spender] == 0. To decrement
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   * @param spender The address which will spend the funds.
   * @param subtractedValue The amount of tokens to decrease the allowance by.
   */
  function decreaseAllowance(
    address spender,
    uint256 subtractedValue
  )
    public
    returns (bool)
  {
    require(spender != address(0));

    _allowed[msg.sender][spender] = (
      _allowed[msg.sender][spender].sub(subtractedValue));
    emit Approval(msg.sender, spender, _allowed[msg.sender][spender]);
    return true;
  }

  /**
   * @dev Internal function that mints an amount of the token and assigns it to
   * an account. This encapsulates the modification of balances such that the
   * proper events are emitted.
   * @param account The account that will receive the created tokens.
   * @param amount The amount that will be created.
   */
  function _mint(address account, uint256 amount) internal {
    require(account != 0);
    _totalSupply = _totalSupply.add(amount);
    _balances[account] = _balances[account].add(amount);
    emit Transfer(address(0), account, amount);
  }

  /**
   * @dev Internal function that burns an amount of the token of a given
   * account.
   * @param account The account whose tokens will be burnt.
   * @param amount The amount that will be burnt.
   */
  function _burn(address account, uint256 amount) internal {
    require(account != 0);
    require(amount <= _balances[account]);

    _totalSupply = _totalSupply.sub(amount);
    _balances[account] = _balances[account].sub(amount);
    emit Transfer(account, address(0), amount);
  }

  /**
   * @dev Internal function that burns an amount of the token of a given
   * account, deducting from the sender's allowance for said account. Uses the
   * internal burn function.
   * @param account The account whose tokens will be burnt.
   * @param amount The amount that will be burnt.
   */
  function _burnFrom(address account, uint256 amount) internal {
    require(amount <= _allowed[account][msg.sender]);

    // Should https://github.com/OpenZeppelin/zeppelin-solidity/issues/707 be accepted,
    // this function needs to emit an event with the updated approval.
    _allowed[account][msg.sender] = _allowed[account][msg.sender].sub(
      amount);
    _burn(account, amount);
  }
}



/**
 * @title FTV
 * @dev Very simple ERC20 Token example, where all tokens are pre-assigned to the creator.
 * Note they can later distribute these tokens as they wish using `transfer` and other
 * `ERC20` functions.
 */
contract FTV is ERC20 {

  string public constant name = "fashion tv";
  string public constant symbol = "FTV";
  uint8 public constant decimals = 8;

  uint256 public constant INITIAL_SUPPLY = 100000000 * (10 ** uint256(decimals));

  /**
   * @dev Constructor that gives msg.sender all of existing tokens.
   */
  constructor() public {
    paused =false;
    ceoAddress = msg.sender;
    cooAddress = msg.sender;
    cfoAddress = msg.sender;
    _mint(msg.sender, INITIAL_SUPPLY);
  }

}